Piecewise Models for Practice Effects in Number Match 3 Response Times
COMPLETED VERSION

SAS Code to Create Slope Pieces:

```
* Example for BALANCED DATA (everyone on same measurement schedule);
* Creating pieces with intercept at session 1, breakpoint at session 2;
* Includes Slope+Deviation and 2 Slopes models;
DATA &udatafile.; SET &udatafile.;
            IF Session = 1 THEN DO; Slope16 = 0; Slope12 = 0; Slope26 = 0; END;
       ELSE IF Session = 2 THEN DO; Slope16 = 1; Slope12 = 1; Slope26 = 0; END;
       ELSE IF Session = 3 THEN DO; Slope16 = 2; Slope12 = 1; Slope26 = 1; END;
       ELSE IF Session = 4 THEN DO; Slope16 = 3; Slope12 = 1; Slope26 = 2; END;
       ELSE IF Session = 5 THEN DO; Slope16 = 4; Slope12 = 1; Slope26 = 3; END;
       ELSE IF Session = 6 THEN DO; Slope16 = 5; Slope12 = 1; Slope26 = 4; END;
       LABEL Slope16  = "Overall Slope from Session 1-6"
             Slope12  = "Early Practice Slope (Session 1-2)"
             Slope26  = "Later Practice Slope (Session 2-6)"; RUN;

* Example for UNBALANCED DATA (uses per-subject time variables instead);
* Creating pieces with intercept at session 1, breakpoint at session 2;
* Includes Slope+Deviation and 2 Slopes models;
DATA &udatafile.; SET &udatafile.;
     IF Session LE 1 THEN DO; Slope16=Time1-Time1; Slope12=Time1-Time1; Slope26=Time2-Time2; END;
ELSE IF Session LE 2 THEN DO; Slope16=Time2-Time1; Slope12=Time2-Time1; Slope26=Time2-Time2; END;
ELSE IF Session LE 3 THEN DO; Slope16=Time3-Time1; Slope12=Time2-Time1; Slope26=Time3-Time2; END;
ELSE IF Session LE 4 THEN DO; Slope16=Time4-Time1; Slope12=Time2-Time1; Slope26=Time4-Time2; END;
ELSE IF Session LE 5 THEN DO; Slope16=Time5-Time1; Slope12=Time2-Time1; Slope26=Time5-Time2; END;
ELSE IF Session LE 6 THEN DO; Slope16=Time6-Time1; Slope12=Time2-Time1; Slope26=Time6-Time2; END;
       LABEL Slope16  = "Overall Slope from Session 1-6"
             Slope12  = "Early Practice Slope (Session 1-2)"
             Slope26  = "Later Practice Slope (Session 2-6)"; RUN;
```

Partial Output from Saturated Means, Unstructured Variances Model for these data (our target):

```
TITLE1 "SAS: Saturated Means, Unstructured Variances Model";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=REML;
       CLASS ID session;
       MODEL nm3rt = session / SOLUTION DDFM=Satterthwaite;
       REPEATED session / R RCORR TYPE=UN SUBJECT=ID;
       LSMEANS session /; RUN;
```

Estimated R Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**301985**	235659	217994	202607	192154	195360
2	235659	**259150**	230217	213232	202092	193268
3	217994	230217	**233368**	205209	196919	188604
4	202607	213232	205209	**217544**	193676	185321
5	192154	202092	196919	193676	**212098**	187840
6	195360	193268	188604	185321	187840	**196733**

Estimated R Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8424	0.8212	0.7905	0.7593	0.8015
2	0.8424	1.0000	0.9361	0.8981	0.8620	0.8559
3	0.8212	0.9361	1.0000	0.9108	0.8851	0.8802
4	0.7905	0.8981	0.9108	1.0000	0.9016	0.8958
5	0.7593	0.8620	0.8851	0.9016	1.0000	0.9196
6	0.8015	0.8559	0.8802	0.8958	0.9196	1.0000

Least Squares Means

Effect	Session #	Estimate	Standard Error	DF	t Value	Pr > |t|
Session	1	1961.89	54.6805	100	35.88	<.0001
Session	2	1815.17	50.6541	100	35.83	<.0001
Session	3	1750.03	48.0684	100	36.41	<.0001
Session	4	1717.80	46.4101	100	37.01	<.0001
Session	5	1707.18	45.8255	100	37.25	<.0001
Session	6	1672.14	44.1345	100	37.89	<.0001

Model 1a. Two Fixed Piecewise Slopes

```
TITLE1 "SAS 2 Fixed Slopes Piecewise Model";
TITLE2 "Random Intercept + VC E";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=REML;
      CLASS ID session;
      MODEL nm3rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT / TYPE=UN SUBJECT=ID;
      REPEATED session / TYPE=VC SUBJECT=ID;
      ESTIMATE "Session 1 Mean" Intercept 1 Slope12 0 Slope26 0;
      ESTIMATE "Session 2 Mean" Intercept 1 Slope12 1 Slope26 0;
      ESTIMATE "Session 3 Mean" Intercept 1 Slope12 1 Slope26 1;
      ESTIMATE "Session 4 Mean" Intercept 1 Slope12 1 Slope26 2;
      ESTIMATE "Session 5 Mean" Intercept 1 Slope12 1 Slope26 3;
      ESTIMATE "Session 6 Mean" Intercept 1 Slope12 1 Slope26 4; RUN;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	ID	202683	29470	6.88	<.0001
Session	ID	34098	2150.11	15.86	<.0001

Fit Statistics

-2 Res Log Likelihood	8382.7
AIC (smaller is better)	8386.7
AICC (smaller is better)	8386.7
BIC (smaller is better)	8391.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1961.89	48.4187	129	40.52	<.0001
Slope12	-163.64	23.2415	503	-7.04	<.0001
Slope26	-32.8932	5.8104	503	-5.66	<.0001

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Session 1 Mean	1961.89	48.4187	129	40.52	<.0001
Session 2 Mean	1798.25	47.0035	115	38.26	<.0001
Session 3 Mean	1765.36	45.9134	104	38.45	<.0001
Session 4 Mean	1732.46	45.5443	101	38.04	<.0001
Session 5 Mean	1699.57	45.9134	104	37.02	<.0001
Session 6 Mean	1666.68	47.0035	115	35.46	<.0001

What is the interpretation of Slope12?

Linear effect of practice over sessions 1-2; for every additional session, RT goes down by 164 ms

What is the interpretation of Slope26?

Linear effect of practice over sessions 2-6: for every additional session, RT goes down by 33 ms

Model 1b. Random Slope 1, Fixed Slope 2

```
TITLE1 "SAS Random Slope 1, Fixed Slope 2 Model";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=REML;
      CLASS ID session;
      MODEL nm3rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT Slope12 / G V VCORR TYPE=UN SUBJECT=ID;
      REPEATED session / TYPE=VC SUBJECT=ID;
RUN;
```

Estimated G Matrix

Row	Effect	Participant ID	Col1	Col2
1	Intercept	101	277818	-69063
2	Slope12	101	-69063	59941

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**301985**	208755	208755	208755	208755	208755
2	208755	**223800**	199632	199632	199632	199632
3	208755	199632	**223800**	199632	199632	199632
4	208755	199632	199632	**223800**	199632	199632
5	208755	199632	199632	199632	**223800**	199632
6	208755	199632	199632	199632	199632	**223800**

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8030	0.8030	0.8030	0.8030	0.8030
2	**0.8030**	1.0000	0.8920	0.8920	0.8920	0.8920
3	0.8030	**0.8920**	1.0000	0.8920	0.8920	0.8920
4	0.8030	0.8920	0.8920	1.0000	0.8920	0.8920
5	0.8030	0.8920	0.8920	0.8920	1.0000	0.8920
6	0.8030	0.8920	0.8920	0.8920	0.8920	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	ID	277818	42741	6.50	<.0001
UN(2,1)	ID	-69063	18932	-3.65	0.0003
UN(2,2)	ID	59941	12743	4.70	<.0001
Session	ID	24168	1702.53	14.20	<.0001

Fit Statistics

-2 Res Log Likelihood	8319.6
AIC (smaller is better)	8327.6
AICC (smaller is better)	8327.7
BIC (smaller is better)	8338.1

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1961.89	54.6805	100	35.88	<.0001
Slope12	-163.64	31.2462	123	-5.24	<.0001
Slope26	-32.8932	4.8916	403	-6.72	<.0001

Is this random first slope, fixed second slope model (1b) better than the both fixed slopes model (1a)? By what criterion?

Yes, REML deviance test: 8383-8320 = 63, which is > 4.6 needed for df=2

Model 1c. Random Slope 1, Random Slope 2

```
TITLE1 "SAS Random Slope 1, Random Slope 2 Model";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=REML;
      CLASS ID session;
      MODEL nm3rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT Slope12 Slope26 / G V VCORR TYPE=UN SUBJECT=ID;
      REPEATED session / TYPE=VC SUBJECT=ID;
RUN;
```

Estimated G Matrix

Row	Effect	Participant ID	Col1	Col2	Col3
1	Intercept	101	284312	-54270	-10644
2	Slope12	101	-54270	63954	-1672.30
3	Slope26	101	-10644	-1672.30	2617.28

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**301985**	230042	219399	208755	198111	187467
2	230042	**257400**	227410	215094	202778	190462
3	219399	227410	**235385**	208013	198314	188615
4	208755	215094	208013	**218604**	193850	186768
5	198111	202778	198314	193850	**207059**	184921
6	187467	190462	188615	186768	184921	**200747**

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8251	0.8229	0.8125	0.7923	0.7614
2	0.8251	1.0000	0.9239	0.9068	0.8784	0.8379
3	0.8229	0.9239	1.0000	0.9170	0.8983	0.8677
4	0.8125	0.9068	0.9170	1.0000	0.9111	0.8916
5	0.7923	0.8784	0.8983	0.9111	1.0000	0.9070
6	0.7614	0.8379	0.8677	0.8916	0.9070	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	ID	284312	42731	6.65	<.0001
UN(2,1)	ID	-54270	18230	-2.98	0.0029
UN(2,2)	ID	63954	13244	4.83	<.0001
UN(3,1)	ID	-10644	3791.26	-2.81	0.0050
UN(3,2)	ID	-1672.30	2097.03	-0.80	0.4252
UN(3,3)	ID	2617.28	636.48	4.11	<.0001
Session	ID	17673	1435.84	12.31	<.0001

Fit Statistics

-2 Res Log Likelihood	8275.4
AIC (smaller is better)	8289.4
AICC (smaller is better)	8289.6
BIC (smaller is better)	8307.7

Construct 95% random effects confidence intervals:

Int = 1962 + 1.96*(284312) → 917 to 3007
Slope12 = -164 + 1.96*(63954) → -659 to 332
Slope26 = -33 + 1.96*(2617) → -133 to 67

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1961.89	54.6805	100	35.88	<.0001
Slope12	-163.64	30.2188	100	-5.42	<.0001
Slope26	-32.8932	6.5888	100	-4.99	<.0001

Is this random both slopes model (1c) better than the random first slope, fixed second slope model (1b)? By what criterion?

Yes, REML deviance test: 8320-8275 = 44, which is > 6.3 needed for df=3

Model 2a. Slope + Deviation Slope

```
TITLE1 "SAS Fixed Slope + Deviation Slope Piecewise Model";
TITLE2 "Random Intercept + VC E";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=REML;
      CLASS ID session;
      MODEL nm3rt = Slope16 Slope26 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT / TYPE=UN SUBJECT=ID;
      REPEATED session / TYPE=VC SUBJECT=ID;
      ESTIMATE "Session 1 Mean" Intercept 1 Slope16 0 Slope26 0;
      ESTIMATE "Session 2 Mean" Intercept 1 Slope16 1 Slope26 0;
      ESTIMATE "Session 3 Mean" Intercept 1 Slope16 2 Slope26 1;
      ESTIMATE "Session 4 Mean" Intercept 1 Slope16 3 Slope26 2;
      ESTIMATE "Session 5 Mean" Intercept 1 Slope16 4 Slope26 3;
      ESTIMATE "Session 6 Mean" Intercept 1 Slope16 5 Slope26 4;
RUN;
```

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	ID	202683	29470	6.88	<.0001
Session	ID	34098	2150.11	15.86	<.0001

Fit Statistics

-2 Res Log Likelihood	8382.7
AIC (smaller is better)	8386.7
AICC (smaller is better)	8386.7
BIC (smaller is better)	8391.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1961.89	48.4187	129	40.52	<.0001
Slope16	-163.64	23.2415	503	-7.04	<.0001
Slope26	130.75	26.6265	503	4.91	<.0001

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
Session 1 Mean	1961.89	48.4187	129	40.52	<.0001
Session 2 Mean	1798.25	47.0035	115	38.26	<.0001
Session 3 Mean	1765.36	45.9134	104	38.45	<.0001
Session 4 Mean	1732.46	45.5443	101	38.04	<.0001
Session 5 Mean	1699.57	45.9134	104	37.02	<.0001
Session 6 Mean	1666.68	47.0035	115	35.46	<.0001

What is the interpretation of Slope16?

Linear effect of practice for all sessions, but based on session 1 to 2: for every additional session, RT goes down by 164 ms

NOW what is the interpretation of Slope26?

Difference in slope from 2 to 6: linear slope from 1 to 6 slows down by 131 ms/session after session 2

Model 2b. Random Slope, Fixed Deviation Slope

```
TITLE1 "SAS Random Slope, Fixed Deviation Slope Model";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=REML;
      CLASS ID session;
      MODEL nm3rt = Slope16 Slope26 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT Slope16 / G V TYPE=UN SUBJECT=ID;
      REPEATED session / TYPE=VC SUBJECT=ID;
RUN;
```

Estimated G Matrix

Row	Effect	Participant ID	Col1	Col2
1	Intercept	101	254290	-12982
2	Slope16	101	-12982	2346.46

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**280225**	241308	228325	215343	202361	189378
2	241308	**256606**	220036	209400	198764	188128
3	228325	220036	**237681**	203457	195168	186878
4	215343	209400	203457	**223449**	191571	185628
5	202361	198764	195168	191571	**213909**	184378
6	189378	188128	186878	185628	184378	**209063**

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8999	0.8847	0.8606	0.8265	0.7824
2	0.8999	1.0000	0.8910	0.8745	0.8484	0.8122
3	0.8847	0.8910	1.0000	0.8828	0.8656	0.8383
4	0.8606	0.8745	0.8828	1.0000	0.8762	0.8588
5	0.8265	0.8484	0.8656	0.8762	1.0000	0.8719
6	0.7824	0.8122	0.8383	0.8588	0.8719	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	ID	254290	37895	6.71	<.0001
UN(2,1)	ID	-12982	3620.52	-3.59	0.0003
UN(2,2)	ID	2346.46	551.40	4.26	<.0001
Session	ID	25934	1827.00	14.20	<.0001

Fit Statistics

-2 Res Log Likelihood	8333.4
AIC (smaller is better)	8341.4
AICC (smaller is better)	8341.4
BIC (smaller is better)	8351.8

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1961.89	52.6735	109	37.25	<.0001
Slope16	-163.64	20.8345	467	-7.85	<.0001
Slope26	130.75	23.2213	403	5.63	<.0001

Is this random slope, fixed deviation slope model (2b) better than the fixed slope, fixed deviation slope model (2a)? By what criterion?

Yes, REML deviance test: 8383-8333 = 49, which is > 4.6 needed for df=2

Model 2c. Random Slope, Random Deviation Slope

```
TITLE1 "SAS Random Slope, Random Deviation Slope Model";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=REML;
      CLASS ID session;
      MODEL nm3rt = Slope16 Slope26 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT Slope16 Slope26 / G V VCORR TYPE=UN SUBJECT=ID;
      REPEATED session / TYPE=VC SUBJECT=ID;
RUN;
```

Estimated G Matrix

Row	Effect	Participant ID	Col1	Col2	Col3
1	Intercept	101	284312	-54270	43626
2	Slope16	101	-54270	63954	-65626
3	Slope26	101	43626	-65626	69916

Estimated V Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**301985**	230042	219399	208755	198111	187467
2	230042	**257400**	227410	215094	202778	190462
3	219399	227410	**235385**	208013	198314	188615
4	208755	215094	208013	**218604**	193850	186768
5	198111	202778	198314	193850	**207059**	184921
6	187467	190462	188615	186768	184921	**200747**

Estimated V Correlation Matrix for ID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8251	0.8229	0.8125	0.7923	0.7614
2	0.8251	1.0000	0.9239	0.9068	0.8784	0.8379
3	0.8229	0.9239	1.0000	0.9170	0.8983	0.8677
4	0.8125	0.9068	0.9170	1.0000	0.9111	0.8916
5	0.7923	0.8784	0.8983	0.9111	1.0000	0.9070
6	0.7614	0.8379	0.8677	0.8916	0.9070	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z
UN(1,1)	ID	284312	42731	6.65	<.0001
UN(2,1)	ID	-54270	18230	-2.98	0.0029
UN(2,2)	ID	63954	13244	4.83	<.0001
UN(3,1)	ID	43626	19049	2.29	0.0220
UN(3,2)	ID	-65626	14154	-4.64	<.0001
UN(3,3)	ID	69916	15434	4.53	<.0001
Session	ID	17673	1435.84	12.31	<.0001

Fit Statistics

-2 Res Log Likelihood	8275.4
AIC (smaller is better)	8289.4
AICC (smaller is better)	8289.6
BIC (smaller is better)	8307.7

Construct 95% random effects confidence intervals:

Int = 1962 + 1.96*(284312) → 917 to 3007
Slope16 = -164 + 1.96*(63954) → -659 to 332
Slope26 = 131 + 1.96*(69916) → -388 to 649

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1961.89	54.6805	100	35.88	<.0001
Slope16	-163.64	30.2188	100	-5.42	<.0001
Slope26	130.75	32.5530	100	4.02	0.0001

Is this random slope, random deviation slope model (2c) better than the random slope, fixed deviation slope model (2b)? By what criterion?

Yes, REML deviance test: 8333-8275 = 58, which is > 6.3 needed for df=3

So which is the better model for these practice effects: polynomials or pieces?
Because these models are non-nested and differ in both their model for the means (fixed effects) and model for the variances (random effects), we need to re-estimate each in ML to compare their fit using AIC and BIC.

```
TITLE1 "SAS: Random Quadratic Model statements";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML;
      CLASS ID session;
      MODEL nm3rt = c1sess c1sess*c1sess / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT c1sess c1sess*c1sess / G V VCORR TYPE=UN SUBJECT=ID;
      REPEATED session / R TYPE=VC SUBJECT=ID; RUN;
```

Fit Statistics	
-2 Log Likelihood	8321.8
AIC (smaller is better)	**8341.8**
AICC (smaller is better)	8342.1
BIC (smaller is better)	**8367.9**

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1945.85	53.5825	101	36.32	<.0001
C1sess	-120.90	19.9481	101	-6.06	<.0001
C1sess*C1sess	13.8656	3.3985	101	4.08	<.0001

Write the random quadratic model equation:

Level 1: $y_{ti} = B_{0i} + B_{1i}(session_{ti} - 1) + B_{2i}(session_{ti} - 1)^2 + e_{ti}$
Level 2: (int) $B_{0i} = \gamma_{00} + U_{0i} \rightarrow 1946 + U_{0i}$
(linear) $B_{1i} = \gamma_{10} + U_{1i} \rightarrow -121 + U_{1i}$
(quad) $B_{2i} = \gamma_{20} + U_{2i} \rightarrow 14 + U_{2i}$

```
TITLE1 "SAS Random Slope 1, Random Slope 2 Model";
PROC MIXED DATA=&udatafile. NOCLPRINT NOITPRINT COVTEST METHOD=ML;
      CLASS ID session;
      MODEL nm3rt = Slope12 Slope26 / SOLUTION DDFM=Satterthwaite;
      RANDOM INTERCEPT Slope12 Slope26 / G V VCORR TYPE=UN SUBJECT=ID;
      REPEATED session / TYPE=VC SUBJECT=ID; RUN;
```

Fit Statistics	
-2 Log Likelihood	8298.9
AIC (smaller is better)	**8318.9**
AICC (smaller is better)	8319.3
BIC (smaller is better)	**8345.1**

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1961.89	54.4091	101	36.06	<.0001
Slope12	-163.64	30.0689	101	-5.44	<.0001
Slope26	-32.8932	6.5561	101	-5.02	<.0001

Write the random two slope model equation:

Level 1: $y_{ti} = B_{0i} + B_{1i}(EarlySlope_{ti}) + B_{2i}(LaterSlope_{ti}) + e_{ti}$
Level 2: (int) $B_{0i} = \gamma_{00} + U_{0i} \rightarrow 1962 + U_{0i}$
(early) $B_{1i} = \gamma_{10} + U_{1i} \rightarrow -164 + U_{1i}$
(later) $B_{2i} = \gamma_{20} + U_{2i} \rightarrow -33 + U_{2i}$

Sample Results Section

Model Specification

Linear mixed models were estimated using SAS PROC MIXED in order to examine the overall pattern of and individual differences in response time over six sessions for a simple processing speed test (number match three). Restricted maximum likelihood (REML) was used in reporting model parameters and to assess the significance of random effects; degrees of freedom were estimated using the Satterthwaite method. The 95% confidence interval (CI) for random variation around each fixed effect was calculated as ± 1.96 standard deviations of its accompanying random variance term.

Although the six sessions were held over a period of 6-10 days, given that experience to the test (and not *time* per se) was the most likely reason for changes in response time, session was used as the metric of time (i.e., as opposed to age or day). Session was centered at the first occasion, such that the intercept represented initial status in all models. Observed mean response times (in milliseconds) estimated from a saturated means model are shown in Figure 1. The intraclass correlation from the unconditional means model (i.e., empty model; intercept only) was calculated as .82, indicating that over 80% of the variance in number match 3 across sessions occurred between persons. Piecewise models were then estimated to approximate the effects of practice across the six sessions, as presented below.

Piecewise Models (note: order of report is different than order of models in handout)

A series of piecewise (i.e., spline) models were also examined in which the effect of session was separated into two slopes: one for the effect of practice from session 1 to 2, and another slope for the effect of practice from session 2 to 6. Models were first specified with a random intercept only. The piecewise model was first parameterized as a deviation model, with one parameter representing the linear slope (i.e., coded as 0-5 for sessions 1- 6), and another parameter representing the deviation from the overall slope in later sessions (i.e., coded 0 for sessions 1-2, and 1,2,3,4 for sessions 3-6). As expected, the coefficient for the deviation slope was significant ($p < .001$), indicating that the slope for sessions 2-6 was significantly shallower than the slope for sessions 1-2.

In order to facilitate interpretation of random effects, the piecewise model was then re-parameterized in order to directly represent the slope between sessions 1-2 (i.e., one parameter coded 0 for session 1, and coded 1 for sessions 2-6), and between sessions 2-6 (i.e., a second parameter coded 0 for sessions 1-2, and coded 1,2,3,4 for sessions 3-6). The improvement to the model from allowing random effects was then evaluated. The addition of a random slope

between sessions 1-2 (as well as a covariance between the random intercept and slope) resulted in a significant improvement to the model, REML deviance difference (2) = 63, $p < .001$. The addition of another random slope between sessions 2-6 (and its two accompanying covariances with the random intercept and session 1-2 slope terms) also resulted in a significant improvement in model fit, REML deviance difference (3) = 44, $p < .001$. The assumption of no residual covariance across sessions after accounting for the random effects was again tested by comparing the fit of a random two-piece model without residual correlations to that of a random two-piece model with residual correlations. Neither a first-order auto-regressive correlation nor a lag-1 correlation resulted in significant improvement in model fit, REML deviance differences (1) < 4, $p > .05$, and thus the random two-piece model without residual correlations across sessions was retained.

The predicted means from the unconditional random two-piece model (i.e., without predictors) are also shown in Figure 1, and model parameters are given in Table 1. The mean predicted response time for the sample at session 1 for the piecewise model was 1962 ms, with a 95% CI of 917 to 3007 ms. The mean rate of change from session 1-2 was -164 ms per session, with a 95% CI of -659 to -332 ms, indicating that most participants were predicted to improve from session 1 to 2. The mean rate of change from session 2-6 was -33 ms per session, with a 95% CI of -133 to 67 ms, indicating that not all participants were predicted to improve further in response times in later sessions.

Comparison of Best-Fitting Unconditional Session Models

The overall fit of the best-fitting unconditional polynomial and piecewise models (see Table 1) were then compared. Because these models differed in both their fixed and random effects and were non-nested, information criteria (in smaller-is-better forms) estimated with maximum likelihood were used to assess differences in fit. The AIC values for the polynomial and piecewise models were 8342 and 8319, respectively, and the BIC values were 8368 and 8345, respectively. Thus, both of the information criteria indicate that the piecewise model appears to have better fit. Additionally, the piecewise model more directly represents the effects of in initial and extended practice across sessions, i.e. by specifying two slopes, rather than slope and deceleration in slope. For these reasons, the unconditional piecewise model with fixed and random effects for slopes from session 1-2 and from session 2-6 was used as a baseline to evaluate the effects of predictors on initial status and rate of improvement in response time across sessions, as presented next.

Table 1

Parameter Estimates and Model Fit Statistics for Number Match 3

Parameter	Piecewise Model		
	Estimate	SE	p-value
Fixed Effects:			
Intercept	1961.9	54.7	< .001
Session 1-2 Slope	-163.6	30.2	< .001
Session 2-6 Slope	-32.9	6.6	< .001
Variance Components:			
Residual Variance	17673	1436	< .001
Intercept Variance	284312	42731	< .001
Intercept-Session 1-2 Covariance	-54270	18230	0.003
Session 1-2 Variance	63954	13244	< .001
Intercept-Session 2-6 Covariance	-10644	3791	0.005
Session 1-2, 2-6 Covariance	-1672	2097	0.425
Session 2-6 Variance	2617	636	< .001
Model Fit:			
REML Deviance	8275		
AIC	8289		
BIC	8308		
Total Number of Parameters	10		

Figure 1

Observed and Model-Predicted Means for Number Match 3